
3DMA, École normale supérieure, PSL University, CNRS, 75005 Paris, France.

louise.leclerc@ens.psl.eu

January 7, 2022

Abstract

The Bochner integral is a generalization of the Lebesgue integral, for functions taking their values
in a Banach space. Therefore, both its mathematical definition and its formalization in the Coq proof
assistant are more challenging as we cannot rely on the properties of real numbers. Our contributions
include an original formalization of simple functions, Bochner integrability defined by a dependent type,
and the construction of the proof of the integrability of measurable functions under mild hypotheses (weak
separability). Then, we define the Bochner integral and prove several theorems, including dominated
convergence and the equivalence with an existing formalization of Lebesgue integral for nonnegative
functions.

1 Introduction

This work is devoted to the Coq formalization of the Bochner integral. Among a huge variety of integrals,
e.g. see [7], the Bochner integral [3] is a generalization of the Lebesgue integral, for real-valued functions,
to the case of functions taking their values in a Banach space, i.e. a complete normed vector space. Thus,
it is perfectly suited for the study of partial differential equations involving time and space variables. For
instance, given a real number T > 0 and a regular enough space domain Ω ⊂ R3, one might be interested in
integrating functions mapping the time interval [0, T ] to the Hilbert space L2(Ω) of functions Ω → R that are
square Lebesgue-integrable, which of course is also a Banach space. In a formal proof setting, it also allows
us to have a single definition and set of theorems for integrating on either R, C, or Rn.

The building of the Bochner integral follows a similar scheme to that of the Lebesgue integral: first
consider simple functions, that only take a finite number of values, define their integral by summing terms of
the form measure of preimage × value, and then extend to the limit of simple functions. The main difference
here is the absence of order in a normed vector space, which prevents the use of monotonicity and of the LUB
property, as in R, and thus prohibits infinite terms in the integral. Instead, it relies on completeness, and
on the additional assumption of separability of the Banach space, or at least of the range of the integrand
function. Note that some mathematical authors and the other formalizations prefer second countability,
which is stronger than separability in general, but actually equivalent in the case of metric spaces (and
Banach spaces are). Rather than the seminal paper by S. Bochner, or the monograph by J. Mikusiński [12],
we chose to follow the modern presentation of the course in real analysis by G. Teschl [15].

The formalization is available at the following link:

https://lipn.univ-paris13.fr/coq-num-analysis/tree/Bochner.1.0/Lebesgue/bochner_integral

where the tag Bochner.1.0 corresponds to the code of this article.

The paper is organized as follows. After drawing up the state of the art in Section 2, Section 3 presents
the Coq formalization of the Lebesgue integral [6] we rely on, and Section 4 some preliminary topological
results. Section 5 is dedicated to simple functions. Bochner integrability is addressed in Section 6, and the
Bochner integral is defined in Section 7. Finally, Section 8 concludes and gives some perspectives.

2 State of the art

Measure theory and nonnegative Lebesgue integration has been formalized in formal proof assistants such as
Mizar1, PVS [14], Isabelle/HOL [13], HOL42, Lean [8, 16], and Coq3. We may cite [2, 9] in Mizar, dedicated
libraries in PVS4, Isabelle/HOL5, and Lean6, [11] in HOL4, and dedicated libraries7 and [6] in Coq.

There are few proof assistants that provide the Bochner integral, while the Riemann or Lebesgue integrals
are more widespread. To the best of our knowledge, there are already two available formalizations.

First, Isabelle/HOL provides Bochner integrability and integral and the dominated convergence theo-
rem [1]. Their goal is probability and the central limit theorem and this generic integral easily encompasses
C and Rn. They assume a second-countable topology (while we add the weaker separability hypothesis at
the only needed point). Their definitions are rather similar to ours except for Bochner integrability: f is
Bochner-integrable if and only if f is measurable and its L1-norm is finite (which is equivalent to saying that
f is absolutely integrable). See our definition in Section 6.

Second and last, Lean provides Bochner integrability and integral, the dominated convergence theorem
and the Fubini theorem [17]. They assume second-countable real Banach space. The main difference is that
the quotient space L1 is defined and that the Bochner integral applies to equivalent classes of functions.
Moreover, the definition of the integral is very different from ours: they extend to L1 the continuous linear
map that is the integral on integrable simple functions.

As a conclusion on this state of the art, the proved theorems are similar, contrary to the definitions.
A difference is that they require second-countability while we locally require weak separability. Another
difference is the simple function definitions: they both rely on the fact that the image is a finite set while we
use a dependent type. See our definition in Section 5.

The present formalization uses the real standard library of Coq, based on [10], and its extension, the
Coquelicot library [4]. These libraries provide support for classical real numbers, which is consistent with the
fact that the mathematics we are formalizing are based on classical logic, as most of the real analysis results
do.

3 The Lebesgue integral in Coq

This work is based on several existing Coq libraries. We of course rely on the standard library, and in
particular for the real numbers [10].

We also rely on Coquelicot [4], which is a conservative extension of the real numbers. We use several
features of this library: the extended real numbers and their operations; the algebraic hierarchy in particular
for the normed modules and Banach spaces; the underlying topology based on filters. We refer the reader to
[4] for more details.

We have also taken inspiration from a recent work defining the Lebesgue integral for nonnegative func-
tions [6]8 and require its latest version. Here are the important design choices of this library. The measur-
ability of subsets of X is formalized as an inductive type parameterized by gen : (X → Prop) → Prop, that
represents the corresponding generated σ-algebra. When X is a metric space, in Coquelicot X : UniformSpace,
one generally uses the Borel σ-algebra that is generated by all the open subsets.

Simple functions are based on lists. More precisely, a function is a simple function when its image is
included in a finite list of values. Then this list may be canonized (by removing unused values and duplicates
and sorting the values) and this canonical list is used to compute the integral of a simple function (provided
a given measure). This cannot be applied here as Banach space values cannot be sorted, contrary to real
numbers.

The integral for nonnegative measurable functions is then defined as in mathematics textbooks:
∫

M+

f dµ = sup
ψ∈SF+

ψ≤f

∫

SF+

ψ dµ

1https://fm.mizar.org/
2https://hol-theorem-prover.org/
3https://coq.inria.fr/refman/
4https://shemesh.larc.nasa.gov/fm/ftp/larc/PVS-library/library/measure_integration.html,

https://shemesh.larc.nasa.gov/fm/ftp/larc/PVS-library/library/lebesgue.html
5https://isabelle.in.tum.de/dist/library/HOL/HOL-Analysis/Measure_Space.html, https://isabelle.in.tum.de/dist/library/HOL/HOL-Ana
6https://leanprover-community.github.io/mathlib_docs/measure_theory/measure/measure_space.html,

https://leanprover-community.github.io/mathlib_docs/measure_theory/integral/lebesgue.html
7https://github.com/math-comp/analysis/
8https://lipn.univ-paris13.fr/MILC/

(a) The full space (in 2D) is separable with the given
points.

(b) The subset Y is heart-shaped. For ensuring its
separability, we may provide points inside Y . But it is
easier to only ensure weak separability by considering
the same points as on the left, that may or may not
belong to Y .

Figure 1: Separability and weak separability: a figurative view.

with SF+ being the set of nonnegative simple functions, and M+ the set of nonnegative measurable functions.
The integral of a nonnegative measurable function f is the supremum of the integral of the nonnegative
measurable simple functions ψ less than or equal to f pointwise. Basic lemmas (such as monotony, scalar
multiplication, addition) are provided in [6], as well the Beppo Levi (monotone convergence) theorem and
Fatou’s lemma.

In the following and for the sake of readability, we will not always specify the scope in the Coq scripts.

4 Some topology in normed modules

We present here some preliminary needed results: a few lemmas above Coquelicot are given in Section 4.1
and separability is described in Section 4.2.

4.1 Additions to Coquelicot

In order to formalize the Bochner integral in Coq, one needs at first some topology in normed vector spaces,
and especially in Banach spaces. In this development, we choose to use the existing formalization of filters and
open subsets of Coquelicot [4]. The main notion required is the limit of sequences, which is straightforwardly
given within Coquelicot. Given a type S : UniformSpace, we may denote

Definition lim_seq (u : nat → S) := lim (filtermap u eventually).

Starting from this definition, one can easily prove its equivalence with the more common textbook defi-
nition, that may also be more practical than filters in some cases.

Lemma is_lim_seq_epsilon {A : AbsRing} {E : NormedModule A} :
∀ u : nat → E, ∀ l : E, is_lim_seq u l ↔
∀ ε, 0 < ε → ∃ N, ∀ n, N ≤ n → ‖ minus (u n) l ‖ < ε.

Another useful lemma we may derive from this notion is the following, stating the (Borel) measurability of a
pointwise limit of measurable functions in any vector space. (Actually, this is not as easy as in the real case,
where one may use LimSup and LimInf to get a simple proof.)

Lemma measurable_fun_lim_seq {X : Set} {gen : (X → Prop) → Prop} :
∀ s : nat → X → E, (∀ n, measurable_fun gen open (s n)) →
∀ f : X → E, (∀ x : X, is_lim_seq (λ n ⇒ s n x) (f x)) → measurable_fun gen open f.

Similarly, we define equivalent Cauchy sequences in a normed vector space, which may be easier to handle
than Cauchy filters in practice:

Definition NM_Cauchy_seq {A : AbsRing} {E : NormedModule A} (u : nat → E) : Prop :=
∀ ε, ε > 0 → ∃ n, ∀ p q, p ≥ n → q ≥ n → ball_norm (u p) ε (u q).

4.2 Separability

Next, we need a formalization of separability in normed vector spaces. Let us remind the mathematical
definition of this property.

Definition 1 (separability) A topological space (E, τ) is said separable when it contains a countable dense
subset, i.e. when there exists a sequence (un)n∈N ∈ EN such that (U is any nonempty open subset)

∀U ∈ τ, U 6= ∅ ⇒ U ∩ {un |n ∈ N} 6= ∅.

It suffices in our case to define a more practical weaker version in which we do not require the countable
part to dwell inside the separable one.

Definition 2 (weak separability) Let (E, τ) be a topological space. A subset Y ⊆ E is said weakly
separable in E when there exists a sequence (un)n∈N ∈ EN such that

∀U ∈ τ, U ∩ Y 6= ∅ ⇒ U ∩ {un |n ∈ N} 6= ∅.

For example, let E be R equipped with the usual topology, Y be R \ Q ⊆ E and (un)n∈N be a sequence
whose range is exactly Q. Then Y (with the induced topology) and (un)n∈N does not satisfy the first definition
since (un)n∈N is not a sequence in Y , but we may say from the second definition that Y is weakly separable
in E, thus avoiding the building of a sequence of irrationals in Y . A more visual and figurative example is
given in Figure 1.

For normed modules, the norm induces the topology, therefore we get the following characterization,
easier to formalize.

Lemma 3 (weak separability in normed vector spaces) Let (E, ‖ · ‖) be a normed vector space. A
subset Y ⊆ E is weakly separable in E if and only if there exists a sequence (un)n∈N ∈ EN such that

∀y ∈ Y, ∀ε > 0, ∃n ∈ N, ‖y − un‖ < ε.

Given E:NormedModule R_AbsRing, we define it in Coq as

Definition NM_seq_separable_weak (u : nat → E) (P : E → Prop) : Prop :=
∀ x : E, P x → ∀ ε : posreal, ∃ n, ball_norm x ε (u n).

Note that the sequence u is explicit in this definition.
For instance, Coq real numbers are weak separable.

Lemma NM_seq_separable_weakR :
NM_seq_separable_weak (λ n ⇒ Q2R (bij_NQ n)) (λ _ : R_NormedModule ⇒ True).

The sequence u ranges over the rationals, relying on the bijection bij_NQ from N onto Q.

5 Formalizing simple functions

A first important step towards Bochner integrability and integral is the definition of simple functions on
a Banach space. Even if the Lebesgue integral also needs simple functions [6], their formalization is not
applicable in our case and we have provided an original definition described in Section 5.1, as well as the
Bochner-integrability. The value of the integral is given in Section 5.2.

5.1 Definition and properties

Following [15], we start by formalizing simple functions. Later in Section 7, we will define the Bochner
integral as a limit of integrals of simple functions (as for the Lebesgue integral).

Let us consider a measurable space (X,Σ), and E a normed vector space that is assumed to be equipped
with its Borel σ-algebra (generated by all open subsets). In Coq, we have X : Set, the σ-algebra Σ is
represented by some generator gen : (X → Prop) → Prop (see Section 3), E : NormedModule A with A : AbsRing,
and its Borel σ-algebra is generated by the generic open : (E → Prop) → Prop. Then, the mathematical
definition of (measurable) simple function is the following.

Definition 4 (simple function) A function f : X → E is said simple when its range is finite and all the
preimages are measurable.

In [6], as explained in Section 3, the simple functions for the Lebesgue integral were defined by the
existence of a list that collects the values taken by the function. This was chosen because by forcing the list
to be sorted, and not to contain any duplicates or unnecessary value, one gets a canonical representation of a
simple function. However, this is no longer possible with vector-valued functions, where no order can be used
on the image space. But it is known that Definition 4 is equivalent to the two following characterizations.

f : X

x

N

0

1

2

which(x)

max_which

E

val(0)

val(1) = val(2)

val (which(x))

0

which val

Figure 2: Coq representation of simple functions.
Given a cutting of the set X into max_which+1 pairwise disjoint measurable parts, the function which : X → nat

maps elements of each part to a distinct index, an integer in the range [[0, max_which]]. Then, the function
val : nat → E maps each index of the previous range to some vector value in E. The greatest index max_which

is mapped to zero. The represented simple function f : X → E is actually the composition of which and val.
Note that several indices (here 1 and 2) may be mapped to the same vector value v (here val 1 equals val 2),
meaning that the preimage f−1({v}) is actually the (disjoint) union of the parts mapped to the indices.
Thus, the representation is not unique. Of course, parts need not be convex, nor connected, and some of
them may be empty, including the last one, associated with the value zero.

Lemma 5 (characterization 1) A function f : X → E is simple if and only if it is a linear combination
of characteristic functions of measurable subsets.

Lemma 6 (characterization 2) A function f : X → E is simple if and only if there exists a finite partition
(Ai)i∈I of X such that for all i ∈ I, f is constant over Ai, and Ai is measurable.

We tried to formalize both in Coq and the second proved to be much more efficient to handle.
The chosen data structure takes the following form.

Record simpl_fun := mk_simpl_fun {
which : X → nat;
val : nat → E;
max_which : nat;
ax_val_max_which : val max_which = zero;
ax_which_max_which : ∀ x : X, which x ≤ max_which;
ax_measurable : ∀ n : nat, n ≤ max_which → measurable gen (λ x ⇒ which x = n);

}.

Such a record tells us that in order to build a simple function, there are three values that should be
given to Coq: which, val, and max_which, see Figure 2, and several proofs. The function which corresponds
to a cutting of the space X , or to an index in the finite set of preimages. The integer max_which is the
maximal value allowed for which (it ensures the finiteness of the cutting). The function val provides the
value corresponding to a given integer.

For instance, suppose we want to construct in Coq the simple function corresponding to f : X → E with
the finite partition (Ai)i∈I of Lemma 6. First of all, because I is finite, we can suppose that it is of the form
[[0, n]] for some n ∈ N (implicitly, here we also suppose that X is not empty), then:

• the number of parts minus one in our cutting (i.e. |I| − 1 or n in the above description) will be stored
into max_which;

• because (Ai)i∈I forms a pairwise disjoint cover of X , for each x ∈ X , there exists a unique i ∈ I such
that x ∈ Ai. The function which will associate this i ∈ N with each x ∈ X . So inside Coq, this becomes
which : X → nat.

• Finally, for every i ∈ I, f takes over Ai a value vi ∈ E. This will be stored inside val i, for every
i : nat. As we will see later on, the value of val for i > n will not matter in our formalization.

In addition to these three values, there is a need for properties that ensure such a structure correctly
represents a simple function and behaves nicely.

• First, we need to ensure that val takes the value zero on max_which. The proof will be stored into
ax_val_max_which. This is not a mathematical consideration but a commodity in order to manipulate
integrability of simple functions. This will allow us to deal with the preimage of 0 separately from the
others, and especially to allow An to be of infinite measure for integrable simple functions.

• Secondly, because which has its values in nat while mathematically, it should have it in [[0, max_which]],
we must ensure that which does not exceed max_which. The proof will be stored into ax_which_max_which.

• Finally, we have to ensure the measurability of the simple function, i.e. the measurability of all its
preimages. The proof will be stored into ax_measurable.

It is convenient to use the record defining a simple function as a function, so we also define the following
coercion.

Definition fun_sf (sf : simpl_fun) : X → E := λ x ⇒ sf.val (sf.which x).

(* So we may write "sf x" for sf : simpl_fun E gen, and x : X. *)

Coercion fun_sf : simpl_fun >−> Funclass.

For instance, for the characteristic function of a (measurable) subset A, we would have which that returns
0 on A and 1 on ¬A; max_which that is 1; and val(n) that is 1 when n = 0 and 0 elsewhere. All assumptions
hold. Therefore, for x ∈ A, we have sf x = sf.val (sf.which x) = sf.val 0 = 1, and for x ∈ ¬A, we have
sf x = sf.val (sf.which x) = sf.val 1 = 0.

Note that the type simpl_fun actually carries more structure than just the definition of simple functions.
As a consequence, the representation of a simple function f by an instance of the record is not unique. Indeed,
the same value v ∈ E could be associated with several distinct indices, meaning that the actual preimage
f−1({v}) could be represented by several (pairwise disjoint) parts with distinct indices. And of course, this
may occur for the value zero, already associated with the index max_which.

Nevertheless, we may recover usual properties about simple functions. An interesting one is their mea-
surability as defined in [6].

Lemma measurable_fun_sf : ∀ sf : simpl_fun E gen, measurable_fun gen open sf.

We have also explicitly constructed an instance of simpl_fun E gen for the sum, opposite, subtraction,
scalar product, norm or power of simple functions.

As an example, given two simple functions f, g : X → E with respective decomposition (Ai)i∈[[0, n]] and
(Bj)j∈[[0,m]], we get a correct decomposition for f + g with (Ai ∩ Bj)(i, j)∈[[0, n]]×[[0,m]]. To formalize this
decomposition in Coq, we used an explicit bijection between [[0, n]] × [[0, m]] and [[0, (n+ 1) · (m+ 1) − 1]].

We will use in the sequel the following notations:

”sf + sg” := sf_plus sf sg.
”− sg” := sf_scal (opp one) sg.
”sf − sg” := sf_plus sf (sf_scal (opp one) sg).
”a · sf” := sf_scal a sf.
”‖ sf ‖” := sf_norm sf.
”sf ˆ p” := sf_power sf p.

Over these simple functions, one also needs to define the integrability property stating that preimages
have a finite measure, except possibly for that of zero (corresponding at least to the index max_which). This
will allow in Section 5.2 to sum terms of the form measure of preimage × value. For the sake of smoothness,
we require all the parts of index smaller than max_which to have finite measure. It therefore prevents parts
of infinite measure with val n = zero and n < max_which. This is allowed in mathematics but impractical in
formal proofs and moreover, this case may be kept out (see below).

Definition integrable_sf (sf : simpl_fun) :=
∀ n, n < sf.max_which → is_finite (µ (λ x ⇒ sf.which x = n)).

Indeed, this definition is not equivalent to the usual mathematical definition of integrability. Simple
functions whose representation involves zero values for indices n < max_which are not recognized as integrable
when the corresponding preimages have infinite measure. But, both definitions match when we ensure that
the only part associated with the value zero is the last one (with index max_which). And this can be proved
through the following result.

Lemma sf_remove_zeros (sf : simpl_fun E gen) :
{ sf’ : simpl_fun E gen | (∀ x : X, sf x = sf’ x) ∧ (∀ n, n < sf’.max_which → sf’.val n 6= zero) }.

Here, we used a sig from Coq, that is a dependent type which contains an instance of simple_fun E gen

together with a proof of (∀ x : X, sf x = sf’ x) and (∀ n, n < sf’.max_which → sf’.val n 6= zero). So we may
use it to remove unwanted zero values from our structure representing a simple function.

The proof of this proposition is straightforward though tedious: browsing all the values in val, suppressing
the redundant zeros and redefining which in order to have which x = sf ’.max_which each time we have that
sf.val (sf.which x) = zero.

Note also that the previous decomposition for the sum of simple functions maintains the integrability.

5.2 The Bochner integral for simple functions

Now that we have defined simple functions, we are able to define the integral of such functions.
Following Section 5.1, let us now consider a measure space (X,Σ, µ) where µ is a measure on the measur-

able space (X,Σ), and E is now a normed vector space over R, still equipped with its Borel σ-algebra. In Coq,
we have now µ : measure gen and E : NormedModule R_AbsRing. Then, we stick to the following mathematical
definition.

Definition 7 (Bochner integral of simple function) Given an integrable simple function s : X → E,
its Bochner integral (relatively to measure µ on X) is defined by

∫

s dµ :=
∑

v∈E

µ
(

f−1{v}
)

· v,

with the convention ∞ · 0E := 0E.

The former sum is finite according to the definition of integrable simple function, and this may be translated
in our formalization by

Definition BInt_sf (µ : measure gen) (sf : simpl_fun _ gen) : E :=
sum_n (λ n ⇒ scal (real (µ (nth_carrier sf n))) (sf.val n)) (sf.max_which).

where nth_carrier is the preimage defined by

Definition nth_carrier (sf : simpl_fun) (n : nat) : (X → Prop) := λ x ⇒ sf.which x = n.

From this definition we may derive the usual properties of the integral such as linearity,

Lemma BInt_sf_lin {sf sg : simpl_fun E gen} (a b : R) :
integrable_sf µ sf → integrable_sf µ sg →
BInt_sf µ (a · sf + b · sg) = a · (BInt_sf µ sf) + b · (BInt_sf µ sg).

Note that · is the scalar multiplication in the normed vector space. The mathematical proof implies some
factorizations and finite sums inversion. This is basic linear algebra, but has proved slightly tedious. Here,
the main difficulty is to handle the measure that takes values in R, and to manage separately:

• α · v when v ∈ E and α ∈ R (i.e. with Coquelicot formalism, is_finite α);

• +∞ · 0E, which equals 0E by mathematical and Coquelicot conventions.

Another usual property is the triangle inequality.

Lemma norm_Bint_sf_le (sf : simpl_fun E gen) : ‖ BInt_sf µ sf ‖ ≤ BInt_sf µ ‖ sf ‖.

This lemma reduces by definition to the usual triangle inequality for a finite sum.

As explained, our formalization of simple function is not canonical, so it must be proved that the value of
BInt_sf sf only depends on the values taken by sf and not on the cutting we chose to represent this function.
This is stated in the following extensionality lemma:

Lemma BInt_sf_ext {sf sf’ : simpl_fun E gen} :
integrable_sf µ sf → integrable_sf µ sf’ → (∀ x : X, sf x = sf’ x) → BInt_sf µ sf = BInt_sf µ sf’.

6 Bochner-integrable functions

Now we define the integrability of functions, by the means of an approximation by simple functions.
Following Section 5.2, let us still consider a measure space (X,Σ, µ), and E is now a Banach space over R,

still equipped with its Borel σ-algebra. In Coq, this becomes E : CompleteNormedModule R_AbsRing. As in

textbooks, we consider f : X → E as the pointwise limit of a sequence (sn)n∈N of simple functions, and such
that

∫

M+
‖f − sn‖ dµ −→

n→∞
0, with

∫

M+
the Lebesgue integral over nonnegative measurable functions. In

this case it may be proved that the sequence
(∫

sn dµ
)

n∈N
is a Cauchy sequence, thus converges, thanks to

the completeness of E, to a vector of E that we may define as the integral of f .

Therefore, we formally define Bochner-integrable functions as follows.

Definition 8 A function f : X → E is said Bochner-integrable (with regard to µ) when there exists a
sequence (sn)n∈N of integrable simple functions such that

• ∀x ∈ X, sn(x) −→
n→∞

f(x);

•

∫

M+
‖f − sn‖ dµ −→

n→∞
0.

This becomes in Coq

Record Bif {f : X → E} := mk_Bif {
seq : nat → simpl_fun E gen;
ax_notempty : inhabited X;
ax_int : ∀ n, integrable_sf µ (seq n);
ax_lim_pw : ∀ x : X, is_lim_seq (λ n ⇒ seq n x) (f x);
ax_lim_l1 : is_LimSup_seq’ (λ n ⇒ LInt_p µ ‖ f − seq n ‖) 0

}.

A function is therefore Bochner-integrable when there exists such a record with the required values and
proofs.

Once again, this record means that in order to prove that f is a Bochner-integrable function, we need
to provide a sequence seq : nat → simpl_fun E gen of simple functions, and several properties corresponding
to the mathematical requirements. In the above record, we used is_LimSup_seq’ which is a generalization to
R-valued sequences of is_LimSup_seq from Coquelicot that only takes reals.

The hypothesis ax_notempty is artificial. It is due to our will to be equivalent to the Lebesgue integral [6]
that requires a nonempty set for preventing empty lists. A solution would be to convince the authors of [6]
to switch to our simple functions.

We then prove several lemmas. First, a Bochner-integrable function is measurable as it is the pointwise
limit of a sequence of measurable simple functions. Then, we also prove that ‖f‖ is integrable in the sense
of Lebesgue integration of Section 3.

We also define approximating sequences of integrable simple functions for the sum, opposite, subtraction,
scalar product and norm of a Bochner-integrable function. From these proofs and as before, we define useful
notations:

”bf + bg” := Bif_plus bf bg.
”− bf” := Bif_scal (opp one) bf.
”bf − bg” := Bif_plus bf (Bif_scal (opp one) bg).
”a · bf” := Bif_scal a bf.
”‖ bf ‖” := Bif_norm bf.

But such a definition of integrability for vector-valued functions, though easy to use, does not make it
really easy to prove that a given function is integrable so we will look for equivalent properties. First of all,
notice that if a function f : X → E is the pointwise limit of simple functions, say (sn)n∈N, then since every
sn has a finite image, the range of f must be weakly separable according to the previous definition. And
since f must be integrable, we also know that

∫

M+
‖f‖ dµ is finite. Reciprocally, given those two properties,

one may wonder if it is possible to prove that f is Bochner-integrable. The answer is yes, and we may even
construct an explicit sequence of simple functions, which will be useful for Coq to compute the value of the
integral of f . To construct such a sequence, we followed [15]. Note that this requires some attention because
it involves some careful splitting of the range of f .

Lemma Bif_separable_range {f : X → E} {u : nat → E} :
inhabited X → measurable_fun gen open f → NM_seq_separable_weak u (inRange f) →
is_finite (LInt_p µ (λ x : X ⇒ ‖ f ‖ x)) → Bif µ f.

The first consequence of this characterization is that any measurable function f : X → R such that
∫

M+
‖f‖ dµ <∞ is Bochner-integrable, because we already know that R is (weakly) separable.

Lemma R_Bif {f : X → R_NormedModule} :
inhabited X → measurable_fun gen open f → is_finite (LInt_p µ (‖ f ‖)) → Bif µ f.

8



So here we recover exactly the definition of integrability for the Lebesgue integral, which makes both
definitions compatible. But now, if we assume X separable and f continuous, one can prove that the range
of f is separable. For example, we deduce that every continuous function f : Rn → E is Bochner-integrable.

For other cases where our function f seems too complicated to prove easily that its range is separable,
we still have the ability to prove its Bochner integrability by using yet another equivalent property.

Lemma 9 A function f : X → E is Bochner-integrable if and only if

• it is the pointwise limit of a sequence of simple functions (without requiring any integrability);

•

∫

M+
‖f‖ dµ <∞.

Functions which are pointwise limit of simple ones (i.e. which satisfies the first dot above) are said strongly
measurable. This definition has been formalized inside the library too, and we have proved several useful
properties about strongly measurable functions. The most striking example is that any pointwise limit of
strongly measurable functions is still strongly measurable. Such a property will be of great use to prove the
dominated convergence theorem.

7 The Bochner integral

The definition of the Bochner integral is straightforward from the integrability definition. Let us still consider
a measure space (X,Σ, µ), and E a Banach space over R, still equipped with its Borel σ-algebra.

Definition 10 Let f : X → E. Let (sn)n∈N be a sequence of integrable simple functions such that

• ∀x ∈ X, sn(x) −→
n→∞

f(x);

•

∫

M+
‖f − sn‖ dµ −→

n→∞
0.

Then the Bochner integral of f (relatively to measure µ on X) is defined by

∫

f dµ := lim
n→∞

∫

sn dµ.

The Coq definition is quite short because the partial function lim_seq was used, so the convergence of the
sequence do need to be checked while defining BInt However, it was proved as an independent lemma, which
is essential to be able to use the properties of BInt bf as a limit.

Definition BInt {f : X → E} (bf : Bif µ f) := lim_seq (λ n ⇒ BInt_sf µ (seq bf n)).

The first property to ensure is that this definition does neither depend on the chosen sequence (sn)n∈N,
nor on the integrability proof. This is stated as the following extensionality lemma.

Lemma BInt_ext {f f’ : X → E} :
∀ (bf : Bif µ f) (bf’ : Bif µ f ’), (∀ x : X, f x = f’ x) → BInt bf = BInt bf’.

We then prove all the expected properties of the integral such as linearity or the triangular inequality, by
taking the limit of the already proved properties over simple functions.

A larger proof is the equality of BInt and LInt_p for nonnegative real-valued integrable functions, as we
had to prove the equivalence of the two formalizations (of simple functions, of integrability and of integrals).
It makes our library compatible with the one about the Lebesgue integral.

The next lemmas were chosen to ease the main perspective of this work, that is the definition of Bochner
spaces in Coq, which are a generalization of Lp spaces for the Lebesgue integral.

Theorem 11 A function f : X → E is zero µ-almost everywhere if and only if
∫

M+
‖f‖ dµ = 0.

Theorem 12 (dominated convergence) Given a nonnegative integrable function g : X → R, and a
pointwise convergent sequence (fn)n∈N of Bochner-integrable functions such that ∀n ∈ N, ‖fn‖ ≤ g, then
f := (x 7→ limn→∞ fn(x)) is Bochner-integrable, and

lim
n→∞

∫

fn dµ =

∫

f dµ.

8 Conclusion and perspective

We have defined the Bochner integral with a constructive point of view for Bochner integrability. We have
proved that a function is Bochner-integrable (with the constructive dependent type definition) provided it is
the pointwise limit of simple functions and that its range is weakly separable. We have also proved that our
definitions are consistent with those of a Coq formalization of the Lebesgue integral.

Our design choices are twofold. Mathematically, we have conscientiously followed Teschl [15] with a kind
of weak separability instead of (regular) separability. Formally, we have simple functions with an index
function and Bochner-integrable by a dependent type. We have succeeded in proving the common lemmas,
from linearity to dominated convergence so this seems a good basis to build upon.

This opens the way to the formalization of Bochner spaces of strongly measurable functions for which
the p-th power of the norm is Lebesgue integrable. They are a generalization of the usual Lp Lebesgue
spaces, where functions equal almost everywhere are also identified. Such spaces are also Banach spaces for
p ≥ 1. For instance, given a regular enough space domain Ω ⊂ R3, the square-integrable functions Ω → R

form the Hilbert space L2(Ω) since R is a Banach space on which Bochner integration applies. Moreover,
given a real number T > 0, the square-integrable functions from [0, T ] to L2(Ω) also form the Hilbert space
L2([0, T ], L2(Ω)). And eventually, this could be used to apply the Lax–Milgram theorem [5] in the context
of the resolution of some set of partial differential equations.

